ISSN 0252-1075
Research Report No. RR-047

Contributions from

Indian Institute of Tropical Meteorology

A FORTRAN ALGORITHM FOR 2-DIMENSIONAL HARMONIC ANALYSIS

By

M. K. TANDON

PUNE - 411 008

INDIA

NOVEMBER 1991

A FORTRAN Algorithm for 2-Dimensional Harmonic Analysis

M.K.Tandon

Indian Institute of Tropical Meteorology, Pune 411 008, INDIA.

RR-47

Abstract

Harmonic analysis is one of the widely used technique in numerical analysis of data in different fields of mathematical sciences. A FORTRAN algorithm for 2-dimensional harmonic analysis is presented in this study. The FORTRAN routines of this algorithm were coded, developed and tested on ND-560/Cx supermini computer system. This algorithm was successfully used in three types of data.

Key Words : Harmonic, Spectral, Fourier, Variability, ND, Supermini, Grid, Dimensional.

Introduction

Harmonic analysis of any grid point distribution of data has been one of the widely used technique in the field of numerical analysis. This perhaps may be due to the simplicity of the trignometric functions involved in this analysis. The FORTRAN language routines described here are based, with some variants, on Spiegel (1974). The algorithm presented here has been successfully executed with three sets of 2-D grid point data. Variation of the variability of the input data explained by the 2-D spectral distribution for different sets of 2-D harmonics is discussed in the text. This FORTRAN algorithm could be used in a problem depending upon the computational requirement of the problem. Efficient Fourier Transform (EFT) algorithm is used in the computations of harmonics.

Methodology

The Fourier series (harmonic) analysis of any two dimensional (2-D) grid point distribution, $f(x,y)$, requires the distribution to be defined in the 2-D interval $[(-L_x,L_x),(-L_y,L_y)]$ and determined outside this 2-D domain by the following relations.

$$f(x + 2L_x,y) = f(x,y) \qquad (1.A)$$

$$f(x,y + 2L_y) = f(x,y) \qquad (1.B)$$

In other words, the 2-D distribution has 2-D periodicity with $2L_x$ and $2L_y$ as its periods along x and y direction, respectively. 1-D Fourier series or Fourier expansion of $f(x,y)$ (for a fixed direction) is defined by the following equation, Spiegel (1974).

$$f(x,y) = \sum_{m=0} [A^c_m (y) \operatorname{Cos}(m\pi x/L_x) + A^s_m (y) \operatorname{Sin}(m\pi x/L_x)] \qquad (2)$$

where 1-D Fourier expansion coefficients $A^c_m (y)$ and $A^s_m (y)$ are computed using the following equations.

$$A^c_m (y) = (1/L_x) \int_{-L_x}^{L_x} f(x,y) \operatorname{Cos}(m\pi x/L_x)\, dx \qquad (3.A)$$

$$A^s_m (y) = (1/L_x) \int_{-L_x}^{L_x} f(x,y) \operatorname{Sin}(m\pi x/L_x)\, dx \qquad (3.B)$$

We include the second dimension, i.e. y direction, by expressing the functional dependence of the 1-D Fourier expansion coefficients $A^c_m(y)$ and $A^s_m(y)$ as another 1-D Fourier series expansion in the following form.

$$A^c_m(y) = \sum_{n=0} [A^{cc}_{mn} \cos(n\pi y/L_y) + A^{cs}_{mn} \sin(n\pi y/L_y)] \qquad (4.A)$$

$$A^s_m(y) = \sum_{n=0} [A^{sc}_{mn} \cos(n\pi y/L_y) + A^{ss}_{mn} \sin(n\pi y/L_y)] \qquad (4.B)$$

where Fourier expansion coefficients A^{cc}_{mn}, A^{cs}_{mn}, A^{sc}_{mn} and A^{ss}_{mn} are computed using the following relations.

$$A^{cc}_{mn} = (1/L_y) \int_{-L_y}^{L_y} A^c_m(y) \cos(n\pi y/L_y)\, dy \qquad (5.A)$$

$$A^{cs}_{mn} = (1/L_y) \int_{-L_y}^{L_y} A^c_m(y) \sin(n\pi y/L_y)\, dy \qquad (5.B)$$

$$A^{sc}_{mn} = (1/L_y) \int_{-L_y}^{L_y} A^s_m(y) \cos(n\pi y/L_y)\, dy \qquad (5.C)$$

$$A^{ss}_{mn} = (1/L_y) \int_{-L_y}^{L_y} A^s_m(y) \sin(n\pi y/L_y)\, dy \qquad (5.D)$$

Coefficients $A^c_m(y)$ and $A^s_m(y)$ appearing in Eqs.(5.A),(5.B),(5.C), and (5.D) are computed using Eqs.(3.A),(3.B). On subsituting Eqs.(4.A) and (4.B) in Eq.(2) we get 2-D Fourier series representation of any 2-D periodic grid point distribution f(x,y) in the following form.

$$f(x,y) = \sum_{m=0} \sum_{n=0} \Big\{ [A^{cc}_{mn} \mathrm{Cos}(n\pi y/L_y) + A^{cs}_{mn} \mathrm{Sin}(n\pi y/L_y)] \mathrm{Cos}(m\pi x/L_x)$$

$$+ [A^{sc}_{mn} \mathrm{Cos}(n\pi y/L_y) + A^{ss}_{mn} \mathrm{Sin}(n\pi y/L_y)] \mathrm{Sin}(m\pi x/L_x) \Big\} \qquad (6)$$

For numerical computations we discretise the continuous features. Let N_x and N_y be the number of grid points along x and y direction, respectively. If N (= N_x or = N_y) is even then the number of cosine (sine) expansion coefficients are N/2 +1 (N/2 - 1). However, if N is odd then the number of cosine [sine] expansion coefficients are (N-1)/2 + 1 [(N-1)/2]. To facilitate simplicity in FORTRAN-77 coding, we introduce (i) for even N, two additional sine expansion coefficients of zero values as the first and $(N/2)^{th}$ sine coefficients and (ii) for odd N, we introduce an additional sine expansion coefficient of zero value as the first sine Fourier expansion coefficient. In FORTRAN-77 computations of these additions are incorporated by prescribing the maximum number of truncations (coefficients) NCX and NCY as per the following (FORTRAN-77) statements

$$NCX = N_x/2 + 1 \qquad (7.A)$$

$$NCY = N_y/2 + 1 \qquad (7.B)$$

for both types (even/odd) of values of N_x and N_y as INTEGER arithmatic of FORTRAN-77 will yield desired values of truncations for both the directions. The discretised equivalent of continuous 2-D Fourier series representation is used in numerical computations and it is given by the following equation.

$$f_{p,q} = \sum_{m=1}^{NCX} \sum_{n=1}^{NCY} \Big\{ [A^{cc}_{mn} \, Cos(2nq\pi/n_y) + A^{cs}_{mn} \, Sin(2nq\pi/n_y)] Cos(2mp\pi/n_x)$$

$$+ [A^{sc}_{mn} \, Cos(2nq\pi/n_y) + A^{ss}_{mn} \, Sin(2nq\pi/n_y)] Sin(2mp\pi/n_x) \Big\} \quad (8)$$

Here (i) NCX and NCY are the maximum values of the truncations along x and y directions, respectively (ii) $f_{p,q}$ is any $(p,q)^{th}$ grid point of the 2-D grid point distribution f(x,y) and $f_{pq} = f(p\Delta x, q\Delta y)$ where Δx (Δy) is the grid length along x (y) direction (iii) n_x (n_y) = $N_x - 1$ ($N_y - 1$) are the number of grid intervals along x (y) direction, respectively and (iv) first point in the discrete representation of Eq.(8) is chosen as the origin.

Eqs.(7.A) and (7.B) give maximum permissible values of truncations for each direction. However, the values of NCX and NCY to be used in FORTRAN code will depend upon the variance (variability) requirements. For any given 2-D periodic grid point distribution, we can tabulate the percentage variance of the grid point distribution explained by the spectral distribution for different number of harmonics as per the different values of the truncations (NCX,NCY). Analysing such a tabular information, we can fix the values of parameters NCX and NCY as per the required acceptable variabilty of the spectral distribution. Variance of a 2-D grid point distribution f(x,y) is computed using the following relations.

$$VAR[f(x,y)] = (N_x N_y)^{-1} [\sum_{i=1}^{N_x} \sum_{j=1}^{N_y} (f_{i,j} - f^*)^2] \quad (9)$$

where f^* is mean value of the distribution f(x,y) and is evaluated as under.

$$f^* = (N_x N_y)^{-1} [\sum_{i=1}^{N_x} \sum_{j=1}^{N_y} f_{i,j}] \quad (10)$$

Percentage variance of the variance, VAR[f], of the original grid point distribution f(x,y) explained by the variance, $VAR[f_s]$, of the spectral distribution is obtained using the following relation.

$$PRCNT = 100 * VAR[f_s] (VAR[f])^{-1} \quad (11)$$

Examples

The algorithm worked out in the earlier section is used on three different sets of 2-dimensional grid point distribution of INPUT data.In each of these three cases we have considered $N_x = N_y = 11$ which gives maximum number of 2-D harmonics in each direction as 6 i.e. maximum value of variables NCX and NCY in the FORTRAN code is 6.The 2-D grid point distribution obtained by using the maximum number of 2-D harmonics, for each direction, and the percentage variance of the INPUT data explained by the spectral OUTPUT data for a set of different numbers of 2-D harmonics is shown in APPENDIX-B for the three cases investigated.In each of these cases it is found that INPUT/OUTPUT grid point distributions exactly tally with each other whenever we consider all 2-D harmonics in the execution of the FORTRAN code.The FORTRAN code for this algorithm is listed in APPENDIX-A. From the tabular information about the percentage variances given in APPENDIX-B, it is found that the percentage variance in each of the three cases increases as (i) value of NCX increases for any fixed value of NCY and (ii) value of NCY increases for any fixed value of NCX. From the tabular information it is also seen that value of the percentage variance for a particular choice of (NCX,NCY) varies from case to case.It is so because the values of 2-D harmonics are intrinsically related with the variability and builtin characteristics of the INPUT 2-D grid point distribution. However, depending upon the nature of the computations required in a problem, the computed 2-D harmonics can further be analysed as per the computational requirements of the problem.

Acknowledgements

Author thanks the Director,I.I.T.M., for providing facilities for carrying out this work. Author also thanks Dr.S.K.Mishra for discussions. Author also thanks Dr.S.Rajamani and Dr.(Mrs.) P.S.Salvekar for reviewing this paper. Utilisation of IITM's ND-560/Cx supermini computer system is thankfully acknowledged.

References

Spiegel, Murray, R., 1974, Fourier Analysis: McGraw-Hill Book Company, pp 191.

APPENDIX-A

LISTING OF ALGORITHM - ROUTINES

..... DESCRIPTION OF VARIABLES .....

... SCALAR VARIABLES ...

```
PI     :  3.1415926536
NX     :  NUMBER OF GRID POINTS ALONG X-DIRECTION.
NY     :  NUMBER OF GRID POINTS ALONG Y-DIRECTION.
NCX    :  VALUE OF TRUNCATION ALONG X-DIRECTION.
NCY    :  VALUE OF TRUNCATION ALONG Y-DIRECTION.

NDM1   :  AN INTERMEDIATE VARIABLE.
NDM2   :  AN INTERMEDIATE VARIABLE.
NDM3   :  AN INTERMEDIATE VARIABLE.
NXHFP  :  AN INTERMEDIATE VARIABLE.
NYHFP  :  AN INTERMEDIATE VARIABLE.
DANGX  :  AN INTERMEDIATE VARIABLE.
DANGY  :  AN INTERMEDIATE VARIABLE.
```

... ARRAY VARIABLES ...

```
BCFSX           :  AN ARRAY STORING ANGLES ASSOCIATED WITH
                   COSINE TERM OF 1-D FOURIER SERIES ALONG
                   X-DIRECTION.

BSFSX           :  AN ARRAY STORING ANGLES ASSOCIATED WITH
                   SINE TERM OF 1-D FOURIER SERIES ALONG
                   X-DIRECTION.

BCFSY           :  AN ARRAY STORING ANGLES ASSOCIATED WITH
                   COSINE TERM OF 1-D FOURIER SERIES ALONG
                   Y-DIRECTION.

BSFSY           :  AN ARRAY STORING ANGLES ASSOCIATED WITH
                   SINE TERM OF 1-D FOURIER SERIES ALONG
                   Y-DIRECTION.

BCSFSX(M,N,1) :  AN ARRAY STORING SUMMATION OF ANGLES
                   ASSOCIATED WITH COSINE TERM OF 1-D
                   FOURIER SERIES ALONG X-DIRECTION.

BCSFSX(M,N,2) :  AN ARRAY STORING SUMMATION OF ANGLES
                   ASSOCIATED WITH SINE TERM OF 1-D
                   FOURIER SERIES ALONG X-DIRECTION.

BCSFSY(M,N,1) :  AN ARRAY STORING SUMMATION OF ANGLES
                   ASSOCIATED WITH COSINE TERM OF 1-D
                   FOURIER SERIES ALONG Y-DIRECTION.

BCSFSY(M,N,2) :  AN ARRAY STORING SUMMATION OF ANGLES
                   ASSOCIATED WITH SINE TERM OF 1-D
                   FOURIER SERIES ALONG Y-DIRECTION.
```

..... DESCRIPTION OF ROUTINES .....

```
CONST  :  IT DEFINES INTERMEDIATE VARIABLES AND COMPUTES
```

```
              ARRAY VARIABLES BCFSX , BSFSX , BCFSY , BSFSY.

    ANGLES  :  IT COMPUTES ARRAY VARIABLES ASSOCIATED WITH 1-D
               FOURIER SERIES.

    SUMANG  :  IT COMPUTES ARRAY VARIABLES BCSFSX , BCSFSY.

    CFSDBL  :  IT COMPUTES FOUR 2-D HARMONICS FROM 2-D GRID POINT
               DISTRIBUTION.

    FORIER  :  IT COMPUTES 1-D HARMONICS FROM 1-D GRID POINT
               DISTRIBUTION.

    SUMDBL  :  IT GENERATES 2-D GRID POINT DISTRIBUTION USING
               FOUR 2-D HARMONICS COMPUTED FROM ROUTINE CFSDBL.

    OBSERV  :  IT COMPUTES VARIANCE OF A 2-D GRID POINT
               DISTRIBUTION.

            .....  DESCRIPTION OF COMMON BLOCKS   .....

      COMMON PI,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,NCX,NCY,DANGX,DANGY,
     &       DX,DY
      COMMON/TO/BCFSX(11),BSFSX(11),BCFSY(11),BSFSY(11),BCSFSX(21,11,2),
     *BCSFSY(21,11,2)

------------------------------------------------------------------------

      SUBROUTINE CONST
      COMMON PI,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,NCX,NCY,DANGX,DANGY,
     &       DX,DY
      COMMON/TO/BCFSX(11),BSFSX(11),BCFSY(11),BSFSY(11),BCSFSX(21,11,2),
     *BCSFSY(21,11,2)
      PI=3.1415926536
      NCX=NX/2+1
      NCY=NY/2+1
      NYHFP=NY/2+1
      NXHFP=NX/2+1
      CALL ANGLES(NCX,PI,NX,BCFSX,BSFSX)
      CALL ANGLES(NCY,PI,NY,BCFSY,BSFSY)
      CALL SUMANG
      RETURN
      END
      SUBROUTINE ANGLES(NCX,PI,NX,BCFS,BSFS)
      DIMENSION BCFS(NCX),BSFS(NCX)
      T=2.*PI/FLOAT(NX)
      C=COS(T)
      S=SIN(T)
      C1=1.
      S1=0.
      DO 1 L=1,NCX
      BCFS(L)=C1
      BSFS(L)=S1
      A=C1*C-S1*S
      B=S1*C+C1*S
      C1=A
    1 S1=B
      RETURN
      END
      SUBROUTINE SUMANG
      COMMON PI,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,NCX,NCY,DANGX,DANGY,
     &       DX,DY
```

```
      COMMON/TO/BCFSX(11),BSFSX(11),BCFSY(11),BSFSY(11),BCSFSX(21,11,2),
     *BCSFSY(21,11,2)
      T=2.*PI/FLOAT(NX)
      ANG=0.
      DO 1 M=1,NXHFP
      ANG=ANG+T
      IF(M.EQ.NXHFP)ANG=T*FLOAT(NX)
      C=COS(ANG)
      S=SIN(ANG)
      C1=1.
      S1=0.
      DO 1 N=1,NCX
      BCSFSX(M,N,1)=C1
      BCSFSX(M,N,2)=S1
      A=C1*C-S1*S
      B=S1*C+C1*S
      C1=A
    1 S1=B
      T=2.*PI/FLOAT(NY)
      ANG=0.
      DO 2 M=1,NYHFP
      ANG=ANG+T
      IF(M.EQ.NYHFP) ANG=T*FLOAT(NY)
      C=COS(ANG)
      S=SIN(ANG)
      C1=1.
      S1=0.
      DO 2 N=1,NCY
      BCSFSY(M,N,1)=C1
      BCSFSY(M,N,2)=S1
      A=C1*C-S1*S
      B=S1*C+C1*S
      C1=A
    2 S1=B
      RETURN
      END
      SUBROUTINE CFSDBL(F,CC,CS,SC,SS)
C
C     THIS ROUTINE SPLITS A 2-D GRID POINT DISTRIBUTION INTO A DOUBLE
C        FOURIES SERIES & COMPUTES 2-D FOURIER EXPANSION COEFFICIENTS
C        CC(M,N),CS(M,N),SC(M,N) AND SS(M,N) USING FOLLOWING SERIES :-
C
C                 NCX  NCY
C                 ---  ---
C                 \    \       CC(M,N)COS(MKX)COS(NLY) + CS(M,N)COS(MKX)SIN(NLY)
C     F(X,Y)=
C                 /    /     + SC(M,N)SIN(MKX)COS(NLY) + SS(M,N)SIN(MKX)SIN(NLY)
C                 ---  ---
C                 M=1  N=1
C
C
      DIMENSION F(21,21),CC(11,11),CS(11,11),SC(11,11),SS(11,11),FC(21),
     *FS(21),FCC(11),FCS(11),FSC(11),FSS(11),G(21),DC(21,11),DS(21,11)
      COMMON PI,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,NCX,NCY,DANGX,DANGY,
     &       DX,DY
      COMMON/TO/BCFSX(11),BSFSX(11),BCFSY(11),BSFSY(11),BCSFSX(21,11,2),
     *BCSFSY(21,11,2)
      DO 5 J=1,NX
      DO 1 K=1,NY
    1 G(K)=F(J,K)
      CALL FORIER(G,FCC,FSS,NCY,NY,BCFSY,BSFSY)
      DO 5 K=1,NCY
```

```
      DC(J,K)=FCC(K)
    5 DS(J,K)=FSS(K)
      DO 7 K=1,NCY
      DO 6 J=1,NX
      FC(J)=DC(J,K)
    6 FS(J)=DS(J,K)
      CALL FORIER(FC,FCC,FCS,NCX,NX,BCFSX,BSFSX)
      CALL FORIER(FS,FSC,FSS,NCX,NX,BCFSX,BSFSX)
      DO 7 J=1,NCX
      CC(J,K)=FCC(J)
      CS(J,K)=FCS(J)
      SC(J,K)=FSC(J)
    7 SS(J,K)=FSS(J)
      RETURN
      END
      SUBROUTINE FORIER(X,Y,Z,NC,NDATA,BCFS,BSFS)
C
C     FIRST DATA POINT IS CHOSEN AS ORIGIN.
C     THIS ROUTINE COMPUTES EXPANSION COEFFICIENTS FOR 1-D GENERAL
C            SERIES.
C
      DIMENSION X(NDATA),Y(NC),Z(NC),BCFS(NC),BSFS(NC)
      MAXFC=NDATA/2+1
      NDM1=NDATA-1
      NDM2=NDATA-2
      NDM3=NDATA-3
      AL=NDATA
      FFACR=2./AL
      CSUM=X(1)
      DO 1 I=2,NDATA
    1 CSUM=CSUM+X(I)
      CSUM=CSUM/AL
      DO 2 I=1,NDATA
    2 X(I)=X(I)-CSUM
      Y(1)=CSUM
      Z(1)=0.
      DO 8 M=2,NC
      C1=BCFS(M)
      S1=BSFS(M)
      UKP2=X(NDM1)
      UKP1=X(NDM2)+(C1+C1)*UKP2
      K=NDM2
      DO 7 I=1,NDM3
      K=K-1
      UK=X(K)+(C1+C1)*UKP1-UKP2
      UKP2=UKP1
    7 UKP1=UK
      Z(M)=FFACR*UK*S1
    8 Y(M)=FFACR*(X(NDATA)+UKP1*C1-UKP2)
      IF((MOD(NDATA,2).EQ.0).AND.(NC.EQ.MAXFC)) Y(NC)=.5*Y(NC)
      RETURN
      END
      SUBROUTINE SUMDBL(A,B,C,D,F)
C
C     THIS ROUTINE SUMS UP A 2-D FOURIES SERIES.
C     HERE (A,B,C,D) ARE SAME AS (CC,CS,SC,SS) CF "CFSDBL".
C
      DIMENSION A(11,11),B(11,11),C(11,11),D(11,11),F(21,21),G(21,11),H(
     *21,11)
      COMMON PI,NX,NY,NDM1,NDM2,NDM3,NXHFP,NYHFP,NCX,NCY,DANGX,DANGY,
     &       DX,DY
      COMMON/TO/BCFSX(11),BSFSX(11),BCFSY(11),BSFSY(11),BCSFSX(21,11,2),
```

```
     *BCSFSY(21,11,2)
      DO 2 N=1,NCY
      DO 2 MX=1,NXHFP
      S=0.
      T=0.
      U=0.
      V=0.
      DO 1 M=1,NCX
      U=U+D(M,N)*BCSFSX(MX,M,2)
      V=V+C(M,N)*BCSFSX(MX,M,1)
      S=S+B(M,N)*BCSFSX(MX,M,2)
    1 T=T+A(M,N)*BCSFSX(MX,M,1)
      IF(MX-NXHFP) 5,6,6
    5 G(NX-MX,N)=T-S
      H(NX-MX,N)=V-U
      G(MX,N)=T+S
      H(MX,N)=V+U
      GO TO 2
    6 G(NX,N)=T+S
      H(NX,N)=V+U
    2 CONTINUE
      DO 3 MX=1,NX
      DO 3 MY=1,NYHFP
      S=0.
      T=0.
      DO 4 M=1,NCY
      S=S+H(MX,M)*BCSFSY(MY,M,2)
    4 T=T+G(MX,M)*BCSFSY(MY,M,1)
      IF(MY-NYHFP) 7,8,8
    7 F(MX,NY-MY)=T-S
      F(MX,MY)=T+S
      GO TO 3
    8 F(MX,NY)=T+S
    3 CONTINUE
      RETURN
      END
      SUBROUTINE OBSERV(F,NX,NY,VAROB)
C
C     VARIANCE COMPUTATIONS FROM GRID POINT DATA.
C
      DIMENSION F(NX,NY)
      S=0.
      DO 1 M=1,NX
      DO 1 N=1,NY
    1 S=S+F(M,N)
      S=S/(FLOAT(NX)*FLOAT(NY))
      VAROB=0.
      DO 2 M=1,NX
      DO 2 N=1,NY
    2 VAROB=VAROB+(F(M,N)-S)*(F(M,N)-S)
      VAROB=VAROB/(FLOAT(NX)*FLOAT(NY))
      RETURN
      END
```

APPENDIX B

EXAMPLES

Case (1)

2-D grid point distribution of INPUT data for this case is ...

... x-Direction ...

0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
0.0	3.45	5.59	5.59	3.45	0.0	-3.45	-5.59	-5.59	-3.45	0.0
0.0	5.59	9.05	9.05	5.59	0.0	-5.59	-9.05	-9.05	-5.59	0.0
0.0	5.59	9.05	9.05	5.59	0.0	-5.59	-9.05	-9.05	-5.59	0.0
0.0	3.45	5.59	5.59	3.45	0.0	-3.45	-5.59	-5.59	-3.45	0.0
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
0.0	-3.45	-5.59	-5.59	-3.45	0.0	3.45	5.59	5.59	3.45	0.0
0.0	-5.59	-9.05	-9.05	-5.59	0.0	5.59	9.05	9.05	5.59	0.0
0.0	-5.59	-9.05	-9.05	-5.59	0.0	5.59	9.05	9.05	5.59	0.0
0.0	-3.45	-5.59	-5.59	-3.45	0.0	3.45	5.59	5.59	3.45	0.0
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0

Percentage variance of the input distribution explained by the spectral distribution for different sets of 2-D harmonics.

NCX ...	1	2	3	4	5	6
NCY						
6	0.0	95.96	98.33	99.12	99.61	100.00
5	0.0	95.58	97.94	98.73	99.21	99.61

4	0.0	95.11	97.46	98.25	98.73	99.12
3	0.0	94.35	96.68	97.46	97.94	98.33
2	0.0	92.08	94.35	95.11	95.58	95.96
1	0.0	0.0	0.0	0.0	0.0	0.0

Grid point distribution obtained from 2-D Fourier series with NCX = 6 and NCY = 6

... x-Direction ...

0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
0.0	3.45	5.59	5.59	3.45	0.0	-3.45	-5.59	-5.59	-3.45	0.0
0.0	5.59	9.05	9.05	5.59	0.0	-5.59	-9.05	-9.05	-5.59	0.0
0.0	5.59	9.05	9.05	5.59	0.0	-5.59	-9.05	-9.05	-5.59	0.0
0.0	3.45	5.59	5.59	3.45	0.0	-3.45	-5.59	-5.59	-3.45	0.0
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
0.0	-3.45	-5.59	-5.59	-3.45	0.0	3.45	5.59	5.59	3.45	0.0
0.0	-5.59	-9.05	-9.05	-5.59	0.0	5.59	9.05	9.05	5.59	0.0
0.0	-5.59	-9.05	-9.05	-5.59	0.0	5.59	9.05	9.05	5.59	0.0
0.0	-3.45	-5.59	-5.59	-3.45	0.0	3.45	5.59	5.59	3.45	0.0
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0

Case (2)

2-D grid point distribution of input data for this case is ...

... x-Direction ...

0.0	5.88	9.51	9.51	5.88	0.0	-5.88	-9.51	-9.51.	-5.88	0.0

0.0	4.76	7.69	7.69	4.76	0.0	-4.76	-7.69	-7.69	-4.76	0.0
0.0	1.82	2.94	2.94	1.82	0.0	-1.82	-2.94	-2.94	-1.82	0.0
0.0	-1.82	-2.94	-2.94	-1.82	0.0	1.82	2.94	2.94	1.82	0.0
0.0	-4.76	-7.69	-7.69	-4.76	0.0	4.76	7.69	7.69	4.76	0.0
0.0	-5.88	-9.51	-9.51	-5.88	0.0	5.88	9.51	9.51	5.88	0.0
0.0	-4.76	-7.69	-7.69	-4.76	0.0	4.76	7.69	7.69	4.76	0.0
0.0	-1.82	-2.94	-2.94	-1.82	0.0	1.82	2.94	2.94	1.82	0.0
0.0	1.82	2.94	2.94	1.82	0.0	-1.82	-2.94	-2.94	-1.82	0.0
0.0	4.76	7.69	7.69	4.76	0.0	-4.76	-7.69	-7.69	-4.76	0.0
0.0	5.88	9.51	9.51	5.88	0.0	-5.88	-9.51	-9.51	-5.88	0.0

Percentage variance of the input distribution explained by the spectral distribution for different sets of 2-D harmonics.

NCX ...	1	2	3	4	5	6
NCY						
6	0.0	95.96	98.33	99.12	99.61	100.00
5	0.0	95.96	98.33	99.12	99.61	100.00
4	0.0	95.95	98.32	99.11	99.60	99.99
3	0.0	95.90	98.27	99.06	99.54	99.54
2	0.0	95.41	97.77	98.56	99.04	99.43
1	0.0	1.45	1.49	1.50	1.51	1.52

Grid point distribution obtained from 2-D Fourier series with NCX = 6 and NCY = 6

... x-Direction ...

0.0	5.88	9.51	9.51	5.88	0.0	-5.88	-9.51	-9.51.	-5.88	0.0

0.0	4.76	7.69	7.69	4.76	0.0	-4.76	-7.69	-7.69	-4.76	0.0
0.0	1.82	2.94	2.94	1.82	0.0	-1.82	-2.94	-2.94	-1.82	0.0
0.0	-1.82	-2.94	-2.94	-1.82	0.0	1.82	2.94	2.94	1.82	0.0
0.0	-4.76	-7.69	-7.69	-4.76	0.0	4.76	7.69	7.69	4.76	0.0
0.0	-5.88	-9.51	-9.51	-5.88	0.0	5.88	9.51	9.51	5.88	0.0
0.0	-4.76	-7.69	-7.69	-4.76	0.0	4.76	7.69	7.69	4.76	0.0
0.0	-1.82	-2.94	-2.94	-1.82	0.0	1.82	2.94	2.94	1.82	0.0
0.0	1.82	2.94	2.94	1.82	0.0	-1.82	-2.94	-2.94	-1.82	0.0
0.0	4.76	7.69	7.69	4.76	0.0	-4.76	-7.69	-7.69	-4.76	0.0
0.0	5.88	9.51	9.51	5.88	0.0	-5.88	-9.51	-9.51	-5.88	0.0

Case (3)

2-D grid point distribution of INPUT data for this case is ...

... x-Direction ...

0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
5.88	4.76	1.82	-1.82	-4.76	-5.88	-4.76	-1.82	1.82	4.76	5.88
9.51	7.69	2.94	-2.94	-7.69	-9.51	-7.69	-2.94	2.94	7.69	9.51
9.51	7.69	2.94	-2.94	-7.69	-9.51	-7.69	-2.94	2.94	7.69	9.51
5.88	4.76	1.82	-1.82	-4.76	-5.88	-4.76	-1.82	1.82	4.76	5.88
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.00
-5.88	-4.76	-1.82	1.82	4.76	5.88	4.76	1.82	-1.82	-4.76	-5.88
-9.51	-7.69	-2.94	2.94	7.69	9.51	7.69	2.94	-2.94	-7.69	-9.51
-9.51	-7.69	-2.94	2.94	7.69	9.51	7.69	2.94	-2.94	-7.69	-9.51
-5.88	-4.76	-1.82	1.82	4.76	5.88	4.76	1.82	-1.82	-4.76	-5.88
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.00

Percentage variance of the input distribution explained by the spectral distribution for different sets of 2-D harmonics.

NCY \ NCX ...	1	2	3	4	5	6
6	1.52	99.43	99.94	99.99	100.00	100.00
5	1.51	99.04	99.54	99.60	99.61	99.61
4	1.50	98.56	99.06	99.11	99.12	99.12
3	1.49	97.77	98.27	98.32	99.33	98.33
2	1.45	95.41	95.90	95.95	95.96	95.96
1	0.0	0.0	0.0	0.0	0.0	0.0

Grid point distribution obtained from 2-D Fourier series with NCX = 6 and NCY = 6

... x-Direction ...

0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
5.88	4.76	1.82	-1.82	-4.76	-5.88	-4.76	-1.82	1.82	4.76	5.88
9.51	7.69	2.94	-2.94	-7.69	-9.51	-7.69	-2.94	2.94	7.69	9.51
9.51	7.69	2.94	-2.94	-7.69	-9.51	-7.69	-2.94	2.94	7.69	9.51
5.88	4.76	1.82	-1.82	-4.76	-5.88	-4.76	-1.82	1.82	4.76	5.88
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.00
-5.88	-4.76	-1.82	1.82	4.76	5.88	4.76	1.82	-1.82	-4.76	-5.88
-9.51	-7.69	-2.94	2.94	7.69	9.51	7.69	2.94	-2.94	-7.69	-9.51
-9.51	-7.69	-2.94	2.94	7.69	9.51	7.69	2.94	-2.94	-7.69	-9.51
-5.88	-4.76	-1.82	1.82	4.76	5.88	4.76	1.82	-1.82	-4.76	-5.88
0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.00
